NONLINEAR ALLPASS LADDER FILTERS IN FAUST

*Julius O. Smith and Romain Michon**

Center for Computer Research in Music and Acoustics
(CCRMA) Stanford University
Palo Alto, CA 94305, USA
jos@ccrma.stanford.edu

ABSTRACT

Passive nonlinear filters provide a rich source of evolving spectra for sound synthesis. This paper describes a nonlinear allpass filter of arbitrary order based on the normalized ladder filter. It is expressed in FAUST recursively in only two statements. Toward the synthesis of cymbals and gongs, it was used to make nonlinear waveguide meshes and feedback-delay-network reverberators.

1. INTRODUCTION

Many musical instruments have important nonlinear effects influencing their sound. In particular, cymbals and gongs exhibit evolving spectra due to nonlinear coupling among their resonant modes [1]. One effective method for efficiently synthesizing such sounds is using the digital waveguide mesh [2, 3] terminated by nonlinear allpass filters [4]. The mesh models linear wave propagation in 2D, while the nonlinear allpass provides nonlinear coupling of the modes of vibration in a way that conserves signal energy, and therefore does not affect damping (which is introduced via lowpass filters at selected points in the mesh). Thus, nonlinear allpass filters provide a valuable tool for nonlinear mode combination while preserving stability and keeping decay-time separately controllable.

2. PASSIVE NONLINEAR FILTERS

2.1. First-Order Switching Allpass

The passive nonlinear filter described in [4] was based on the idea of terminating a vibrating string on two different springs k_1 and k_2, as shown in Fig. 1. The switching spring-constant creates a nonlinearity in the string-spring system. Importantly, the switching from one spring to the other only accurs when the spring displacements are zero, so that energy is not affected. (The potential energy stored in a spring k_i displaced by x_i meters is given by $k_i x_i^2/2$ [5].)

In an ideal vibrating string with wave impedance R [5], terminating the string by an ideal spring k_i provides an *allpass reflectance* at the end of the string for traveling waves. That is, reflected displacement waves $y^-(t)$ at the termination are related to the incident waves $y^+(t)$ by

$$Y^-(s) = Y^+(s)H_i(s)$$

* CCRMA visiting researcher from Saint Étienne University, France, supported by the ASTREE Project



Figure 1: *Vibrating string terminated by two different springs k_1 and k_2. Only one spring is active at a time.*

where $H_i(s)$ is the (Laplace-domain) transfer function of the allpass filter

$$H_i(s) = \frac{s - k_i/R}{s + k_i/R}.$$

Replacing the ideal string by a digital waveguide [5] and digitizing the spring reflectance $H_i(s)$ via the bilinear transform yields the digital reflectance

$$H_i(z) = -\frac{a_i + z^{-1}}{1 + a_i z^{-1}}, \qquad a_i = \frac{k_i - 2Rf_s}{k_i + 2Rf_s}$$

where f_s denotes the sampling rate in Hz. While the digital reflectance remains an allpass filter due to properties of the bilinear transform, energy conservation is only approximately obtained, except when the allpass state variable happens to be exactly zero when the coefficient a_i is switched from a_1 to a_2 or vice versa.

2.2. Delay-Line Length Modulation

Since allpass filters are fully characterized by their time-delay at each frequency, the switching allpass of the previous section can be regarded as a form of nonlinear delay-line length modulation in which the delay line switches between two allpass-interpolated lengths (different at each frequency in general).

Delay-line length modulation has been used previously to simulate nonlinear string behavior. For example, the length modulations due to tension variations have been addressed [6]. Additionally, it has long been recognized that the highly audible nonlinearity of the sitar is due to the continuous length modulation caused by its curved bridge [1]. Similarly, the tambura nonlinearly modulates its string length between two lengths via a cotton thread near the bridge [1]. In digital waveguide models such as `Sitar.cpp` in the Synthesis ToolKit (STK) [7], delay-line length is modulated without careful regard for energy conservation; this normally works out fine in practice because lengthening a delay-line is energy conserving when the new samples are zero, and shortening the delay-line is typically a bit lossy and never energy-creating.

3. NONLINEAR ALLPASSES OF ARBITRARY ORDER

We propose to extend the nonlinear switching allpass in two ways:

1. Any order allpass can be used (not just first order).
2. Any kind of coefficient modulation can be used (not just switching between two values at zero crossings of some state variable).

Our method is based on the Normalized Ladder Filter (NLF) [8]. Such filters can be derived from digital waveguide filters by using normalized traveling waves in place of ordinary physical traveling waves [5], where the normalization is chosen so that the square of the traveling-wave amplitude equals the power associated with that sample throughout the waveguide network.

Figure 2 shows the first-order NLF allpass as it is typically drawn [9, 5].



Figure 2: *First-order normalized ladder allpass filter, with coefficients $c = \cos(\theta)$ and $s = \sin(\theta)$, $\theta \in [-\pi, \pi]$.*

Figure 3 shows the same filter as it is depicted in the block diagram rendered by "`faust -svg`" for the FAUST expression

```
process =
  _ <: *(s),(*(c):(+:_)~(*(-s))):_, mem*c:+;
```

The function `allpassnn(1)` is equivalent to this in the FAUST distribution (`filter.lib` after 2/2/11).



Figure 3: *First-order normalized ladder allpass filter as drawn by `faust -svg`.*

A general property of allpass filters is that each delay element can be replaced by an allpass to produce another allpass (a unit-circle to unit-circle conformal map of the transfer function). If the delay element in Fig. 2 is replaced by itself at the output of another first-order allpass of the same form (Fig. 2), then a second-order NLF allpass is obtained. This process is extended to arbitrary orders by replacing the rightmost delay by a first-order NLF allpass followed by a delay. This same recursive construction works also for the Kelly-Lochbaum ladder allpass, and the two-multiply and one-multiply lattice structures [9, 5]. It was used to show equivalence of the two-multiply lattice structure to direct-form-II first-order nested allpass filters in [10, 9].

In FAUST, NLF allpass filters of arbitrary order are conveniently specified by means of the pattern-matching facility:

```
allpassnn(0,tv) = _;
allpassnn(n,tv) = _ <: *(s), (*(c) :
  (+ : allpassnn(n-1,tv))~(*(-s)))
  : _,mem*c:+
  with {
    c=cos(take(n,tv));  s=sin(take(n,tv));
  };
```

This is the full definition of `allpassnn()` in `filter.lib`.

Figure 4 shows the block diagram generated for the second-order NLF allpass specified as `allpassnn(2,tv)`.

4. ENERGY INVARIANCE ILLUSTRATION

The following FAUST program illustrates the energy-invariance property of the normalized-ladder allpass for order 3:

```
import("filter.lib"); // for allpassnn
N = 3; // allpass filter order
theta = par(i,N,PI*noise@(i+1));
rms = ^(2) : smooth(0.999) : sqrt;
uvwnoise = sqrt(3)*noise; // unit variance
process =  uvwnoise <: _, allpassnn(N,theta)
           : rms, rms;
```

The input and output rms levels plotted by `faust2octave` are virtually indistinguishable by eye, as shown in Figure 5. Thus, when applied to unit-variance white noise, the output signal (and in fact every state-variable within the filter [8]) is unit-variance white noise, even as the filter coefficients are modulated over their maximum range by uncorrelated white noise.

5. APPLICATIONS

5.1. Nonlinear Waveguide Mesh

A simple example terminating a square waveguide mesh with nonlinear allpass filters is shown in Figures 6 and 7, generated from the following FAUST source:

```
import("effect.lib"); // for mesh_square()
nlmesh(N,NA,x)=mesh_square(N)~(apbank(4*N,x))
with {
  coeffs(x)=par(i,NA,x); // e.g.
  apbranch(i,x) = allpassnn(NA,coeffs(x),x);
  apbank(M,x) = bus(M)
    : par(i,M-1,apbranch(i)),
                apbranch(M-1) + x;
};
N=2;  // mesh order (nonnegative power of 2)
NA=1; // allpass order (any positive integer)
process = nlmesh(N,NA);
```

The input signal is added into a corner of the mesh, where all modes are excited.



Figure 4: *Second-order NLF allpass as drawn by* `faust -svg`*.*



Figure 5: *RMS level of input and output signals for the third-order normalized-ladder allpass driven by white noise, with uncorrelated white noise used as reflection coefficients. The two curves are nearly indistinguishable.*

5.2. Nonlinear Feedback Delay Network

It was reasoned that the higher-order nonlinear allpass might enable structures simpler than a full waveguide mesh for simulating nonlinearly coupled modes. Thus, another test along these lines was to insert the nonlinear allpass into each lane of a Feedback Delay Network (FDN) reverberator (trivially modifying `fdnrev0` in `effect.lib`). The nonlinear "reverberator" so obtain was then tested by listening to its impulse response. While simple cymbals were not obtainable for the cases tried, some very nice metallic-plate synthesis was obtained, especially when using small delay-line lengths in the FDN reverb. An interesting sonic phenomenon obtained was a rising perceptual pitch as the impulse density was increased. In other words, a faster "drum roll" has a higher spectral centroid. A valuable performable dimension is the degree of nonlinearity, implemented by a scale factor on the reflection coefficients. (When the reflection coefficients are all zero, the allpass reduces to a pure delay.)



Figure 7: *Nonlinear allpass bank showing mesh corner excitation.*

5.3. Nonlinear Tubes and Strings

In a companion paper [11], we report on applications of the nonlinear allpass to FAUST implementations of digital waveguide synthesis instruments. Especially nice results were obtained for the clarinet and harpsichord.

6. CONCLUSIONS

The nonlinear allpass based on a recursively defined normalized ladder filter was found to be useful for the nonlinear synthesis of metallic plates by terminating a waveguide mesh on its rim and



Figure 6: 2×2 *square waveguide mesh terminated on nonlinear allpass filters around its rim.*

by inserting it in the feedback paths of an FDN reverberator. By varying the degree of nonlinearity, a useful performance dimension is obtained.

7. REFERENCES

[1] N. H. Fletcher and T. D. Rossing, *The Physics of Musical Instruments, 2nd Edition*, Springer Verlag, New York, 1998.

[2] S. A. Van Duyne and J. O. Smith III, "Physical modeling with the 2-D digital waveguide mesh," Tokyo, Japan, Sept. 1993, pp. 40–47.

[3] L. Savioja and V. Välimäki, "Reducing the dispersion error in the digital waveguide mesh using interpolation and frequency-warping techniques," vol. 8, no. 2, pp. 184–194, Mar. 2000.

[4] J. R. Pierce and S. A. Van Duyne, "A passive nonlinear digital filter design which facilitates physics-based sound synthesis of highly nonlinear musical instruments," vol. 101, no. 2, pp. 1120–1126, Feb. 1997.

[5] J. O. Smith III, *Physical Audio Signal Processing*, `https://ccrma.stanford.edu/~jos/pasp/`, Dec. 2010, online book.

[6] T. Tolonen, V. Välimäki, and M. Karjalainen, "Modeling of tension modulation nonlinearity in plucked strings," vol. SAP-8, pp. 300–310, May 2000.

[7] P. Cook and G. Scavone, *Synthesis ToolKit in C++*, `http://ccrma.stanford.edu/software/stk/`, 2010.

[8] A. H. Gray and J. D. Markel, "A normalized digital filter structure," *IEEE Trans. Acoustics, Speech, Signal Processing*, vol. ASSP-23, no. 3, pp. 268–277, June 1975.

[9] J. D. Markel and A. H. Gray, *Linear Prediction of Speech*, Springer Verlag, New York, 1976.

[10] D. Massie, "An engineering study of the four-multiply normalized ladder filter," vol. 41, no. 7/8, pp. 564–582, 1993.

[11] R. Michon and J. O. Smith, "FAUST-STK: A set of linear and nonlinear physical models for the FAUST programming language," in *in review*, 2011.

[12] M. Karjalainen, J. Backman, and J. Pölkki, "Analysis, modeling, and real-time sound synthesis of the kantele, a traditional Finnish string instrument," New York, 1993, pp. 229–232, IEEE Press.